	ATL TRANSFORMATION EXAMPLE	
	Book to Publication	Date 18/03/2005

1. ATL Transformation Example

1.1. Example: Book → Publication

The Book to Publication example describes a very simple transformation task. In the metamodel Book the class Book contains an ordered set of Chapters. These Chapters hold the information of the number of pages of Chapters. The metamodel Publication is simpler; its class Publication contains a title and the total number of pages. For the transformation, all chapters of a Book have to be visited to calculate the sum of all pages.

1.1.1. Metamodels

The source metamodel Book (see Figure 1 Book) consists of the class Book and the class Chapter. Each Book has a title and a set of chapters. Each Chapter has a title and a number of pages.




Figure 1 Book

The target metamodel Publication (see Figure 2 Publication) consists of the class Publication which holds a title and the number of pages.





Figure 2 Publication

	ATL TRANSFORMATION EXAMPLE	
	Book to Publication	Date 18/03/2005

1.1.2. Rules Specification

These are the rules to transform a Book model to a Publication model:

- For each Book instance, a Publication instance has to be created. The attributes of the Publication instance are set as follows:
 - The title of a Publication has to be set with the title of a Book.
 - The total number of pages of a Publication is the sum of the pages of the Chapters of a Book.

1.1.3. ATL Code for the Example Book → Publication

The ATL code for the transformation of a Book model to a Publication model consists of one rule, Book2Publication. This rule uses a helper in which all pages of all Chapters of a Book are being added up. In fact the sum can be calculated by adding up in iterations (*getNbPages*) or by using the sum operation (*getSumPages*).

```

module Book2Publication;
create OUT : Publication from IN : Book;

-- getNbPages collects the nbPages of all chapters
-- and calculates the sum

helper context Book!Book def : getNbPages() : Integer =
    self.chapters->collect(f|f.nbPages)->
        iterate(pages; acc : Integer = 0 |
            acc + pages)
;

-- getSumPages does the same as getNbPages,
-- but it uses the OCL sum operation

helper context Book!Book def : getSumPages() : Integer =
    self.chapters->collect(f|f.nbPages).sum()
;

rule Book2Publication {
    from
        b : Book!Book
    to
        out : Publication!Publication (
            title <- b.title,
            nbPages <- b.getSumPages()
        )
}

```